
Zusammenfassung 

SAS bietet die Möglichkeit, ZIP Dateien direkt einzulesen und zu erstellen. Seit der SAS 
Version 9.2 können ZIP Dateien mit dem ODS Statement ODS PACKAGE erstellt wer-
den. Mit SAS 9.4 kommt die Option hinzu, mit dem „FILENAME ZIP“ Statement ZIP Da-
teien zu erstellen und einzulesen. In diesem Tipp & Trick Beitrag wird gezeigt, wie ZIP 
Dateien mit den beiden aufgeführten Optionen (ODS Statement ODS PACKAGE und 
Statement FILENAME ZIP) genutzt werden können. In Bezug auf das „FILENAME ZIP“ 
Statement wird auch dargestellt, wie man sich eine Inhaltsübersicht anzeigen lassen kann. 
Und zu guter Letzt wird veranschaulicht, wie man mit Hilfe der Applikation 7-Zip, von 
SAS aus, passwortgeschützte ZIP Dateien verarbeitet werden. 

 
Schlüsselwörter: ODS PACKAGE, filename ZIP, 7-ZIP, WinZIP, WinRAR 

1 ODS PACKAGE 

1.1 ODS PACKAGE ZIP – FILENAME ZIP 

In der folgenden Tabelle werden die wichtigsten Eigenschaften des ODS Ausgabekanals 
PACKAGE ZIP und des Statements FILENAME ZIP gegenübergestellt. 
 
Tabelle 1: ODS PACKAGE – FILENAME ZIP 

Eigenschaften ODS PACKAGE ZIP (1) FILENAME ZIP (2) 
Erstellen Ja Ja 
Hinzufügen Ja Ja 
Unterordner Ja Ja 
Alle Dateitypen Ja Ja 
Öffnen (ZIP/GZIP/GZ …) Nein Ja 
Inhalte anzeigen (Ja) Ja 
Passwortgeschützte Dateien Nein Nein 

(1) Verfügbar seit SAS Version 9.2 
(2) Verfügbar seit SAS Version 9.4 

1.2 ODS PACKAGE ZIP – Aufbau 

In der folgenden Tabelle wird der Aufbau des ODS PACKAGE ZIP erläutert. 
 
Tabelle 2: ODS PACKAE ZIP – Aufbau 

ods package(newzip) open nopf; Mit diesem Statement wird der ODS 
PACKAGE Ausgabekanal geöffnet. Die 
NOPF Option unterdrückt, dass zusätzliche 
Metadaten erstellt werden. Standardmäßig ist 
die Option an, die Option fügt einen Paket 
Metadaten Eintrag hinzu, dieser Eintrag kann 
von anderen SAS Applikationen gelesen und 
für die Weiterverarbeitung genutzt werden. 

ods package(newzip) add 
file=datei1 path="datei/"; 

Die <datei1> wir dem Paket hinzugefügt, mit 
der Option PATH kann eine Unterordnerstruk-
tur erstellt werden. In meinem Beispiel wird 
der Ordner <datei> erstellt. 

ods package(newzip) add 
file=datei2; 

Eine zweite Datei <datei2> wird dem Paket 
hinzugefügt. 

ods package(newzip) publish 
archive  

  properties( 

   archive_name="ausgabe.zip"  

   archive_path="C:\temp" 

 ); 

In dem PUBLISH ARCHIVE Statement werden 
die Ausgabeparameter festgelegt, 
archive_name = Name der ZIP Datei + Da-
teiendung, archive_path = Ausgabepfad, 
wo die ZIP Datei gespeichert werden soll. 

ods package(newzip) close; Mit diesem Statement wird der ODS 
PACKAGE Ausgabekanal geschlossen und 
die ZIP Datei wird erstellt. 

1.3 ODS PACKAGE ZIP – Beispiel Programm ZIP Datei erstellen 

Der folgenden SAS Programmcode erstellt zwei Dateien: shoes.csv und cars.rtf. 
 
%let ausgabedir = C:\temp; 
ods listing close; 
filename datei1 "&ausgabedir.\shoes.csv"; 
ods csv file=datei1; 
proc print data=sashelp.shoes(obs=10)noobs; 
 format _all_; 
run; 
filename datei2 "&ausgabedir.\cars.rtf"; 
ods rtf file=datei2; 
proc print data=sashelp.cars(obs=10)noobs; 

 format _all_; 
run; 
ods _all_ close; 
ods listing; 
filename _all_ clear; 

Im nächsten Schritt erstelle ich eine ZIP Datei mit dem Namen ausgabe.zip. Die bei-
den Dateien shoes.csv und cars.rtf aus dem vorherigen Programmcode Beispiel 
werden in die ZIP Datei geschrieben. Die Datei shoes.csv wird in den Unterordner 
datei gespeichert und die Datei cars.rtf auf den ersten Level der ZIP Datei. 

%let ausgabedir = C:\temp; 
ods package(newzip) open nopf; 
ods package(newzip) add file=datei1 path="datei/"; 
ods package(newzip) add file=datei2; 
ods package(newzip) publish archive  
  properties( 
   archive_name="ausgabe.zip"  
   archive_path="&ausgabeDir." 
  ); 
ods package(newzip) close; 

 
Inhalt der Datei 
ausgabe.zip 

 

Abbildung 1: Datei ausgabe.zip Inhalt 

 
Aufbau der Datei 
ausgabe.zip 

 

Abbildung 2: Datei ausgabe.zip Aufbau 

2 Filename ZIP 

2.1 Filename ZIP – ZIP Datei erstellen 

In dem folgenden Beispiel wird eine ZIP Datei mit dem Filename ZIP Statement er-
stellt. 
 

Tabelle 3: Filename ZIP – ZIP Datei erstellen 

filename inData "C:\Program 
Files\SASHome\SASFoundation\9.4
\core\sashelp\class.sas7bdat"; 

Als erstes wird ein Filename Statement zur 
Datei, welche in die ZIP Datei gespeichert 
werden soll, erstellt. 

filename outZip zip 
"c:\temp\sas_dateien.zip" 

Danach folgt ein Filename Statement zur 
ZIP Datei, die erstellt werden soll. 

member='class.sas7bdat'; Der Dateiname der Datei, auf die im Filena-
me Statement inData referenziert wurde, ist 
Teil des Filename ZIP Statements. 

data _null_; 

    infile inData recfm=n; 

    file outZip recfm=n; 

    input byte $char1. @; 

    put byte $char1. @; 

run; 

Die ZIP Datei wird mit Hilfe eines data 
_null_ Schrittes erstellt. 

filename _all_ clear; Die Verknüpfung zwischen SAS und den 
Dateien class.sas7bdat und 
sas_dateien.zip wird gelöst. 

 
Inhalt der Datei 
sas_dateien.zip 

 

Abbildung 3: Datei sas_dateien.zip Inhalt 

2.2 Filename ZIP – Dateien in eine existierende ZIP schreiben 

In diesem Beispiel wird eine existierende ZIP Datei erweitert. 
 
Tabelle 4: Filename ZIP – ZIP Datei erweitern 

filename inData "C:\Program 
Files\SASHome\SASFoundation\9.4
\core\sashelp\cars.sas7bdat"; 

Wie im vorherigen Beispiel, wird als erstes 
ein Filename Statement zur Datei, die der 
ZIP Datei hinzugefügt werden soll, erstellt.  

filename outZip zip 
"c:\temp\sas_dateien.zip" 

Danach folgt ein Filename Statement zur 
vorhandenen ZIP Datei. 

member='unterordner/ 
cars.sas7bdat'; 

Der Dateiname der Datei auf die im Filena-
me Statement inData referenziert wurde ist 
Teil des Filename ZIP Statements. Zusätz-
lich wird in diesem Beispiel ein Unterordner 
erstellt, in dem die SAS Datei 
cars.sas7bdat gespeichert wird. 

data _null_; 

    infile inData recfm=n; 

    file outZip recfm=n; 

    input byte $char1. @; 

    put byte $char1. @; 

run; 

Die SAS Datei cars.sas7bdat wird in die 
vorhandene ZIP Datei sas_dateien.zip 
geschrieben. 

filename _all_ clear; Die Verknüpfung zwischen SAS und den 
Dateien cars.sas7bdat und 
sas_dateien.zip wird gelöst. 

 
Inhalt der Datei 
sas_dateien.zip 

 

Abbildung 4: Datei sas_dateien.zip Inhalt 

2.3 Filename ZIP – ZIP Dateien entpacken 

In dem nächsten Beispiel wird gezeigt wie man mit dem Filename ZIP Statement eine 
ZIP Datei entpackt. 
 
Tabelle 5: Filename ZIP – ZIP Datei entpacken 
filename inZip zip 
"C:\temp\sas_dateien.zip"; 

Im ersten Schritt wird das Filename State-

ment zur ZIP Datei, die entpackt werden, 

erstellt. 
filename outDs 
"%SYSFUNC(pathname(work))/class
.sas7bdat"; 

Danach folgt ein Filename Statement, wohin 
die Datei entpackt werden soll. In diesem 
Beispiel in die SAS Work Bibliothek. 

data _null_; 
  infile inZip(class.sas7bdat)   
  lrecl=256 recfm=F  
  length=length eof=eof unbuf; 
  file outDs lrecl=256 recfm=N; 
  input; 
  put _infile_ $varying256.  
  length; 

  return; 
    eof: 
  stop; 

run; 

Die ZIP Datei sas_dateien.zip wird ent-
packt, der Inhalt class.sas7bdat wird in 
der SAS Work Bibliothek gespeichert. 

filename _all_ clear; Die Verknüpfung zwischen SAS und den 
Dateien class.sas7bdat und 
sas_dateien.zip wird gelöst. 

 
Der entpackte Inhalt der Datei 
sas_dateien.zip 

 

Abbildung 5: SAS Work Bibliothek 

2.4 Filename ZIP – Inhalt einer ZIP Datei anzeigen 

Um das Entpacken einer ZIP Datei zu automatisieren, sollte man den Inhalt kennen. Im 
nächsten Beispiel wird gezeigt, wie man eine Inhaltsangabe von einer ZIP erstellt und 
diese in einer SAS Datei speichert. 
 
Tabelle 6: Filename ZIP – Inhalte anzeigen 
filename inZip zip 
"C:\temp\sas_dateien2.zip"; 

Wie auch bei den anderen Beispielen, 
muss zuerst das Filename Statement 
zur ZIP Datei erstellt werden. 

data work.contents(keep=memname 
isFolder); 

length memname $200 isFolder 8; 
fid=dopen("inzip"); 
   if fid=0 then stop; 
   memcount=dnum(fid); 
   do i=1 to memcount; 
      memname=dread(fid,i); 
      isFolder =  
      (first(reverse(trim 
      (memname)))='/'); 
      output; 
   end; 
   rc=dclose(fid); 
run; 

Die ZIP Datei wird mit dem Befehl 
dopen geöffnet, mit dem Befehl dread 
wird der Inhalt gelesen und der Befehl 
dclose schließt die ZIP Datei wieder. 
Zusätzlich wird überprüft, ob kein füh-
render Slash vorhanden ist. Die In-
haltsübersicht wird dann in die SAS 
Datei work.contents geschrieben. 

filename _all_ clear; Die Verknüpfung zwischen SAS und 
der Datei sas_dateien2.zip wird 
gelöst. 

 
Inhalt der SAS Datei 
contents 

 

Abbildung 6: Datei contents Inhalt 

3 Passwortgeschützte ZIP Datei entpacken 

Passwortgeschützte ZIP Dateien können nicht mit SAS Boardmitteln entpackt werden. 
Dafür werden externe Tools wie z.B. WinZIP, WinRAR oder 7-ZIP benötigt. Im nächs-
ten Beispiel wird dargestellt, wie man passwortgeschützte ZIP Dateien mit 7-ZIP ent-
packt. 
 
Tabelle 7: Passwortgeschützte ZIP Dateien entpacken 

options noxwait noxsync; Da ich 7-ZIP über die Windows 
Command Line aufrufe, sollten  folgende 
SAS System Optionen zu Anfang gesetzt 
werden: noxwait, noxsync 

%let source_path=C:\temp; 
%let source_filename=ausgabe2.zip; 
%let passwd=Ksfe1234!; 

Zur Einfachheit, erstelle ich drei SAS 
Makro Variablen: source_path = Pfad, 
wo die ZIP Datei gespeichert wurde, 
source_filename = Name der ZIP Datei 
und passwd = Das Passwort, um die ZIP 
Datei zu entpacken. 

data _null_; 
   newdir=dcreate(“extract”, 
”&source_path.”); 
run; 

Mit dem Befehl dcreate wird ein neuer 
Ordner erstellt, in diesen wird die ZIP Da-
tei entpackt. 

%SYSEXEC 
“C:\temp\executable\7_zipportable\
App\ 
7-Zip64\7z.exe“ x “&source_path. 
\&source_filename.“ -o 
“&source_path.\extract“ -
p“&passwd.“ ; 

Im nächsten Schritt rufe ich mit dem SAS 
Makro Befehl %SYSEXEC das 7-ZIP 
Command Line Tool auf. Das Command 
Line Tool kann im folgenden 7-ZIP Ord-
ner gefunden werden: App\7-
Zip64\7z.exe 
 
Übersicht der verwendeten Kommandos: 
X = Entpacken, der volle Pfad wird benö-
tigt. 
-O = Ausgabeverzeichnis 
-P = Passworteingabe 

 
Entpackter Inhalt der 
ZIP Datei 
ausgabe2.zip 

 

Abbildung 7: Entpackter Inhalt 

